

FB1 - <offline>

"azyklischeDaten" Auslesen und Schreiben azyklischer Daten über Profibus

Name: KOSTAL **Familie:** PROFIBUS**Autor:** Werner **Version:** 0.1**Bausteinversion:** 1**Zeitstempel Code:** 19.09.2013 15:21:02**Interface:** 19.09.2013 15:21:02**Längen (Baustein / Code / Daten):** 00738 00574 00010

| Name | Datentyp | Adresse | Anfangswert | Kommentar |
|-------------------|----------|---------|---------------|--|
| IN | | 0.0 | | |
| enable | Bool | 0.0 | FALSE | Dient der Aktivierung und Übernahme angelegter Daten. |
| readwrite | Bool | 0.1 | FALSE | False bedeutet Daten lesen und True bedeutet Daten schreiben. |
| diagnosticAddress | DWord | 2.0 | DW#16#0 | Diagnose-Adresse des angesprochenen Profibus-Gerätes in HEX |
| parameterNumber | Int | 6.0 | 0 | Nummer des anzusprechenden Parameters. |
| writeValue | Real | 8.0 | 0.000000e+000 | Übergabewert für schreibenden Zugriff auf Parameter |
| OUT | | 0.0 | | |
| errorParam | Bool | 12.0 | FALSE | Es ist ein Fehler beim Login aufgetreten. |
| errorWrite | Bool | 12.1 | FALSE | Es ist ein Fehler beim Schreiben aufgetreten. |
| errorRead | Bool | 12.2 | FALSE | Es ist ein Fehler beim Lesen aufgetreten. |
| busy | Bool | 12.3 | FALSE | Bearbeitung noch nicht abgeschlossen. |
| readValid | Bool | 12.4 | FALSE | Gelesener Wert ist gültig |
| writeDone | Bool | 12.5 | FALSE | Wert erfolgreich übertragen |
| readValue | DWord | 14.0 | DW#16#0 | Rückgabewert des Parameters beim Zugriff lesend. |
| blockStatus | DWord | 18.0 | DW#16#0 | Rückgabe des Bausteinstatus |
| IN_OUT | | 0.0 | | |
| STAT | | 0.0 | | |
| flagUsed | Bool | 22.0 | FALSE | Flag gibt an, dass der Baustein zur Zeit in Verwendung ist. |
| operation | Bool | 22.1 | FALSE | Gepeicherter Wert für Lesen oder Schreiben in laufender Kommunikation. |
| done_param | Bool | 22.2 | FALSE | Auswahl des Parameters erfolgreich |
| firstCall | Bool | 22.3 | FALSE | Kennzeichnet Phase ! des Aufrufs |
| parameter | DWord | 24.0 | DW#16#0 | Gespeicherter Wert der Parameter-Nummer der laufenden Kommunikation. |
| diagnosticAddr | DWord | 28.0 | DW#16#0 | Gespeicherte Diagnos-Adresse der laufenden Kommunikation |
| write_value | DWord | 32.0 | DW#16#0 | Gespeicherter Wert für Übertragung in der laufenden Kommunikation. |
| TEMP | | 0.0 | | |
| temp_record | DWord | 0.0 | | Momentanwert für den lesenden Zugriff auf einen Parameter |
| temp_len | Int | 4.0 | | Momentanwert der Anzahl übertragener Bytes. |

Baustein: FB1 azyklische Daten**Netzwerk: 1** Ablaufsteuerung des Programmbausteins

```
//Status der Kommunikation abfragen:
CLR
UN  #enable           //Freigabe nicht erfolgt und
UN  #firstCall        //erster Aufruf der laufenden Kommunikation sowie
UN  #flagUsed         //Baustein ist noch beschäftigt, dann
BEB                               //Baustein nicht ausführe.
//Sprung zu Phase 1 der Kommunikation
CLR                               //Initialisiere VKE=0 für nächste Überprüfung
U   #firstCall        //Anmeldungsphase aktiv
SPB  logi             //Springe zur Phase 1 Anmeldung (login)
//Sprung zu Phase 2 der Kommunikation
```

```
CLR                                //Initialisiere VKE=0 für nächste Überprüfung
U    #flagUsed                    //Phase 1 der Kommunikation durchgeführt
UN   #operation                   //Daten Lesen aktiviert
SPB  read                        //Springe zu Phase 2 der Lesen (read)
CLR                                //Initialisiere VKE=0 für nächste Überprüfung
U    #flagUsed                    //Phase 1 der Kommunikation durchgeführt
U    #operation                   //Daten Schreiben aktiviert
SPB  writ                        //Springe zu Phase 2 Schreiben (write)
//Übernahme der Daten für die laufende Kommunikation:
SET
R    #operation
CLR
U    #readwrite                   //Operation übernehmen
=    #operation
L    #writeValue                 //Zu schreibenden Wert übernehmen
T    #write_value
L    #parameterNumber           //Parameternummer übernehmen
T    #parameter
L    #diagnosticAddress         //Diagnoseadresse übernehmen
T    #diagnosticAddr
```

| | |
|-------------|-----------------------------|
| Netzwerk: 2 | Parameter auswählen (login) |
|-------------|-----------------------------|

```
//Kennzeichne Kommunikation als begonnen:
SET
S    #firstCall
//Bausteinaufruf für Parameterauswahl:
logi: CALL "WRREC" , DB53
REQ  :=TRUE
ID   :=#diagnosticAddr
INDEX :=0
LEN  :=4
DONE :=#done_param
BUSY :=#busy
ERROR :=#errorParam
STATUS:=#blockStatus
RECORD:=#parameter
//Der erste Schritt der Kommunikation ist beendet und wird als solche gekennzeichnet.
CLR                                //Initialisiere VKE=0 für nächste Überprüfung
UN   #done_param                  //Die Auswahl des Parameters ist abgeschlossen
BEB
SET
S    #flagUsed                    //Phase 1 der Kommunikation abgeschlossen
R    #firstCall
BEB
```

| | |
|-------------|-----------------|
| Netzwerk: 3 | Daten Schreiben |
|-------------|-----------------|

```
writ: CLR                                //Initialisiere VKE=0 für nächste Überprüfung
U    #errorParam
BEB                                //Bei einem Fehler wird Bausteinausführung hier abgebrochen.

//Aufruf der Systemfunktion zum Schreiben von Daten:
CALL "WRREC" , DB54
REQ  :=#done_param
ID   :=#diagnosticAddr
INDEX :=1
LEN  :=4
DONE :=#writeDone
BUSY :=#busy
ERROR :=#errorWrite
STATUS:=#blockStatus
RECORD:=#write_value
//Ausgangsparameter initialisieren:
CLR                                //Initialisiere VKE=0 für nächste Überprüfung
UN   #writeDone
BEB
SET
R    #flagUsed
BEB
```

| | |
|-------------|-------------|
| Netzwerk: 4 | Daten lesen |
|-------------|-------------|

```
read: CLR
U    #errorParam
BEB                                //Bei einem Fehler wird Bausteinausführung hier abgebrochen.
//Ausgangsparameter initialisieren:

L    DW#16#0                      //Überschreibe bisherige Ausnagsvariable mit Null.
T    #readValue
//Aufruf der Systemfunktion zum lesen der Daten.
CALL "RDREC" , DB52
REQ  :=#done_param
ID   :=#diagnosticAddr
```

```
INDEX :=1
MLEN  :=4
VALID :=#readValid
BUSY  :=#busy
ERROR :=#errorRead
STATUS:=#blockStatus
LEN   :=#temp_len
RECORD:=#temp_record
```

```
//Parameterdaten separieren
```

```
CLR
UN   #readValid
BEB
SET
R    #flagUsed
L    #temp_record
T    #readValue
BE
```